/**
 * @(#) CongFamilyProxy.as
 *
 * @date		2009. 03. 19.
 * @version		1.0
 * @author		Park, Juneho
 * @description 가족정보 Proxy - Proxy
 * 				가족정보를 조회 한다.
 */
package com.skt.common_hr.congrest.model
{
	import com.skt.common_hr.common.Variables;
	import com.skt.common_hr.common.puremvc.CommonProxy;
	import com.skt.common_hr.congrest.CelebrationFacade;
	import com.skt.common_hr.congrest.model.vo.CongFamilyVO;
	import com.wcl.dataloader.*;

	import mx.collections.ArrayCollection;
	import mx.managers.CursorManager;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;

	public class FamilyProxy extends CommonProxy
	{
		public static const NAME:String = "CongFamilyProxy";

		//은행조회
		public var defaultResultMessage:String;

		public var family:ArrayCollection;
		public var familyVo:CongFamilyVO;
		public var familyDefault:CongFamilyVO;


		/**
		 * 생성자
		 *
		 * @param
		 * @return void
		 */
		public function FamilyProxy(data:Object=null)
		{
			super(NAME, data);
		}


		/**
		 * data load를 수행
		 * 		- DataLoader 를 생성하고 Import Parameter 객체를 생성, RFC를 호출
		 *
		 * @param pernr 사번
		 * @return void
		 */
		public function load(pernr:String):void
		{
			var param:Object = new Object();
			param["I_PERNR"] = pernr;

			loadData("Z_HR_0040", param, result, fault);
		}


		public function defaultLoad(pernr:String):void
		{
			var param:Object = new Object();
			param["I_PERNR"] = pernr;

			loadData("Z_HR_0040", param, defaultResult, defaultFault);
		}


		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Event Listeners
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 *
		 * @param
		 * @return void
		 */
		public function result(event:ResultEvent):void
		{
			if(event.result["RETCODE"]==0) {
				family = parseFamilyList(event.result.FTAB);
				trace("loaded family");
				facade.sendNotification(CelebrationFacade.LOAD_FAMILY_COMPLETE, this);
			}
			else {
				trace( "failed to Family");
				facade.sendNotification(CelebrationFacade.LOAD_FAMILY_FAILED, this);
			}
			CursorManager.removeBusyCursor();
		}

		/**
		 * 서버에서 오류가 발생했을 처리
		 *
		 * @param
		 * @return void
		 */
		private function fault(event:FaultEvent):void
		{
			trace(event.fault.faultString);
			facade.sendNotification(CelebrationFacade.LOAD_FAMILY_FAILED, this);
			CursorManager.removeBusyCursor();
		}

		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 *
		 * @param
		 * @return void
		 */
		public function defaultResult(event:ResultEvent):void
		{
			if(event.result["RETCODE"]==0) {
				familyVo = parseFamilyDefault(event.result.FTAB);
				facade.sendNotification(CelebrationFacade.LOAD_FAMILY_INFO, this);
			}
			else {
				trace( "failed to Family");
				defaultResultMessage = event.result["RETTEXT"];
				facade.sendNotification(CelebrationFacade.LOAD_DEFAULT_FAMILY_FAILED, this);
			}
		}

		/**
		 * 서버에서 오류가 발생했을 처리
		 *
		 * @param
		 * @return void
		 */
		private function defaultFault(event:FaultEvent):void
		{
			defaultResultMessage = event.fault["RETTEXT"];
			facade.sendNotification(CelebrationFacade.LOAD_DEFAULT_FAMILY_FAILED, this);
		}




		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Data Parsing functions
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 *
		 * @param
		 * @return 가족관계 Value Object
		 */
		private function parseFamilyList(srcArr:ArrayCollection):ArrayCollection
		{
			var retArr:ArrayCollection = new ArrayCollection();

			for each(var rec:Object in srcArr)
			{
				var vo:CongFamilyVO = new CongFamilyVO();
				vo.KDSVH		= rec["KDSVH"];		//자녀와의 관계
				vo.FAMILY		= rec["FAMILY"];	//가족코드
				vo.ZTEXT		= rec["ZTEXT"];		//경조 가족관계 텍스트
				vo.FNAME		= rec["FNAME"];		//사원 또는 지원자의 포맷된 이름
				//vo.REGNO		= rec["REGNO"];		//주민등록번호 (한국)
				vo.REGNO		= vo.getPid(rec["REGNO"]);
				if(vo.ZTEXT=='') {
					vo.ZTEXT='-';
				}
				retArr.addItem(vo);
			}
			return retArr;
		}

		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 *
		 * @param
		 * @return 관족관계(본인) Value Object
		 */
		private function parseFamilyDefault(srcArr:ArrayCollection):CongFamilyVO
		{
			var retArr:ArrayCollection = new ArrayCollection();

			for each(var rec:Object in srcArr)
			{
				if(rec["FAMILY"] == "01"){
					var vo:CongFamilyVO = new CongFamilyVO();
					vo.KDSVH		= rec["KDSVH"];				//자녀와의 관계
					vo.FAMILY		= rec["FAMILY"];			//가족코드
					vo.ZTEXT		= rec["ZTEXT"];				//경조 가족관계 텍스트
					vo.FNAME		= rec["FNAME"];				//사원 또는 지원자의 포맷된 이름
					vo.REGNO		= vo.getPid(rec["REGNO"]);  //주민등록번호 (한국)
					
					break;
				}
			}
			return vo;
		}


	}
}